

SERIE DE TP #01

Thème : Programmation sous Office via les Macros VBA

Exercice #01 : « Modèle & Formulaire Utilisateur »

Objectifs de la manipulation

- Se familiariser avec l'environnement de développement ;
- Découvrir la notion de projet et ses composantes ;
- Créer un projet de Mini-Calculatrice à travers lequel vous manipulerez un nombre important de fonctionnalités.



On voudrait réaliser un formulaire utilisateur sous Word qui permettra de saisir des fiches de note de contrôle continu puis de les insérer dans un tableau Word qui se trouve dans un document créé sur la base d'un modèle « dotm ».

Nous allons le faire à travers les étapes suivantes :

1. Créer un modèle de document Word que vous enregistrez sous le nom « ModèleNotesCC-Gr<Num de votre groupe>.dotm ». Le but est d'éviter de travailler avec le modèle « Normal.dot » ou « Normal.dotm ».
2. Créer ensuite un document word sur la base du modèle de la question 1 et que vous enregistrerez sous le nom « NotesCC-Gr<Num de votre groupe>.docm ».
3. Dans ce document, placer le contenu suivant :

Liste des Notes de Contrôle Continu

Matricule	Nom	Prénom	Note CC1	Note CC2

4. A partir du ruban « Développeur », placer en bas du tableau un bouton dont le rôle sera d'afficher le formulaire qui devra permettre d'insérer les informations dans le tableau de la question 3. Vous nommerez ce bouton « AjoutLigne » et vous lui changerez son titre vers « Fiche Etudiant » par la propriété « caption ».
5. Basculer vers le visual basic puis créer un formulaire contenant les mêmes champs (avec les mêmes noms) que le tableau de la question 3. Ajouter aussi deux boutons « Ajouter » et « Fermer » sur le formulaire. Voici à quoi doit ressembler le formulaire :

Par la table des propriétés, nommer le formulaire « FicheEtudiant »¹.

6. Dans l'événement « click » du bouton « AjoutLigne » de la question 4 taper dedans l'instruction permettant d'afficher le formulaire. L'instruction est : `FicheEtudiant.show`

Tester le bouton. Que remarquez-vous ? Essayer en suite l'instruction `FicheEtudiant.show(0)` . Qu'en déduisez-vous ?

7. Dans l'événement « click » du bouton « Annuler », placer l'instruction `FicheEtudiant.Hide` puis tester sur le document Word. Qu'en déduisez-vous ?
8. De même, pour le bouton titré « Ajouter »², réaliser les contrôles de chaque champ en interceptant l'événement « EXIT » à l'aide de l'instruction `If ... End If` afin de s'assurer que les champs de type texte ne sont pas vides ou numériques et que les champs des notes sont bien dans l'intervalle [0 ... 20]. En cas de présence d'erreur, afficher un message d'avertissement à l'aide de l'instruction `MsgBox`. Voici deux exemples de code à écrire pour intercepter l'événement EXIT du champ Matricule (chaîne numérique) et du champ Nom (chaîne alphabétique) :

```
Private Sub Matricule_Exit(ByVal Cancel As MSForms.ReturnBoolean)
'
' Contrôler Matricule
'
Matricule.Text = Trim(Matricule.Text)
If Matricule.Text = "" Then 'Matricule vide
    MsgBox "Matricule VIDE !!", vbExclamation, "Erreur !"
    Cancel = True
ElseIf Not IsNumeric(Matricule.Text) Then
    MsgBox "Matricule non NUMERIQUE !!", vbExclamation, "Erreur !"
    Cancel = True
Else
    Nom.SetFocus
End If
End Sub
```

Nom


exit

```
Private Sub Nom_exit(ByVal Cancel As MSForms.ReturnBoolean)
'
' Contrôler Nom
'
Nom.Text = Trim(Nom.Text)
If Nom.Text = "" Then 'Nom vide
    MsgBox "Nom VIDE !!", vbExclamation, "Erreur !"
    Cancel = True
ElseIf Not Nom.Text Like "[A-Za-z]*" Then
    MsgBox "Nom non ALPHABETIQUE !!", vbExclamation, "Erreur !"
    Cancel = True
Else
    Prenom.SetFocus
End If
End Sub
```

Lorsque tout est correct, vous insérerez le contenu de la fiche dans une ligne du tableau en utilisant les instructions suivantes :

<code>Selection.TypeText <nom d'un champs ></code>	<code>//Permet de placer un contenu dans une cellule</code>
<code>Selection.MoveRight</code>	<code>//Aller vers la cellule suivante (à droite)</code>

¹ Le tout collé, i-e, pas d'espace entre les mots.



² La variable associée à ce bouton se nomme save :

Après avoir discuté les avantages et inconvénients de cette solution en séance de TP, nous allons procéder à une seconde approche de programmation du bouton « Ajouter » en manipulant la structure de la table « Liste des Notes » elle-même par programmation, ce qui nous permettra de maîtriser les ajouts dans le tableau et non n'importe où dans le document.

Dans cette seconde solution, la table des notes est traitée comme une matrice de plusieurs lignes et 5 colonnes où il s'agira d'insérer chaque champ de la fiche dans la cellule lui correspondant dans la prochaine ligne vide disponible.

9. Pour le faire, nous allons mettre en commentaire les lignes écrites précédemment puis taper les lignes actives telles que vous le montre cette capture d'écran :

```
Private Sub Save_Click()
'-----
' Première Solution
'-----
' Ajouter les valeurs des champs dans la table
'Selection.TypeText Text:=Matricule.Text
'Selection.MoveRight
'Selection.TypeText Text:=Nom.Text
'Selection.MoveRight
'Selection.typeText Text:=Prenom.Text
'Selection.MoveRight
'Selection.typeText Text:=NoteCC1.Text
'Selection.MoveRight
'Selection.typeText Text:=NoteCC2.Text
'-----
' Seconde Solution
'-----
' Déclarer les variables nécessaires
Dim MyDocument As Document ' pointeur vers document
Dim MyTable As Table
Dim NumLigne As Integer

' Initialiser les variables
Set MyDocument = ThisDocument
Set MyTable = MyDocument.Tables(1)

' Trouver la prochaine ligne vide dans la table
NumLigne = MyTable.Rows.Count
If Trim(MyTable.Cell(NumLigne, 1).Range.Text) <> vbCr Then
    MyTable.Rows.Add
    NumLigne = NumLigne + 1
End If

' Transférer les données depuis la fiche vers la table
MyTable.Cell(NumLigne, 1).Range.Text = Matricule.Value
MyTable.Cell(NumLigne, 2).Range.Text = Nom.Value
MyTable.Cell(NumLigne, 3).Range.Text = Prenom.Value
MyTable.Cell(NumLigne, 4).Range.Text = NoteCC1.Value
MyTable.Cell(NumLigne, 5).Range.Text = NoteCC2.Value

' Vider les champs du formulaire après chaque Ajout
Matricule.Text = ""
Nom.Text = ""
Prenom.Text = ""
NoteCC1.Text = ""
NoteCC2.Text = ""
Matricule.SetFocus

End Sub
```